

Chapter 2

Simple Types

Before jumping straight into dependent types with all its beautiful complexities we will first have a look at something simpler, appropriately named *simple types*. We start of with the most fundamental type of all, the function type $_ \rightarrow _$ (section 2.1). One issue with simple types (which is repeated next chapter when we look at propositional logic) is that it is not clear what to start with. I will bypass this issue by introducing constructions which work for any type, called `Set` in Agda. That is we will be using *polymorphic* constructions without introducing a formal account for polymorphism which we will do later once we have dependent types. A particular example of polymorphic constructions are identity `id` and composition $_ \circ _$ (section 2.2) which give us a first taste of category theory.

We will have a superficial look at λ -calculus (section 2.3), by which I always mean typed λ -calculus because otherwise it doesn't make any sense to me. At least enough to see that variables are causing a bit of a headache, something which can be avoided by using combinatory logic which features just two combinators (`S` and `K`), but the price we have to pay is the creation of some completely unreadable combinator code. However, I think it is worthwhile to know about them and we will return to combinators later from a more formal perspective.


Functions are not all there is to simple types and hence we are going to look at sums ($_ \uplus _$) and products ($_ \times _$) (section 2.5) which let us encode finite types. Finally we will meet some old acquaintances from high school in the context of type theory.

2.1 Functions

In Type Theory functions are a fundamental concept. This is different from set theory where functions are viewed as a special kind of relations, namely relations that assign to each element of the domain (the input) exactly one element of the codomain (the output). In Type Theory it is the other way around: we start with functions and actually we need functions to say what a relations is,

namely a relation between A and B is a function from A and B to the type of propositions. But lets not get ahead of ourselves.

Intuitively a function between types A and B is a black box where we can input elements of $a : A$ and out come elements of $f a : B$.



You may notice that we don't write $f(a)$ as usual in Mathematics and in many programming languages but instead save brackets and just write $f a$, which is common in functional programming languages like Haskell. We write $A \rightarrow B$ for the type of functions from A to B and hence we write $f : A \rightarrow B$ to express that f is such a function.

I was saying *black box*, even though for reasons of readability I didn't actually draw a black box. The meaning here is rather metaphorical, it means we cannot look into the box. If somebody gives you a function, all you can do it is to feed it elements and observe the output. You have no way to explore the mechanism (this is a sort of digital rights management for functions). This aspect will become important when we discuss extensionality later.

Let's look at some examples! For the purpose of illustration I will use some types like the natural numbers $\mathbb{N} = \{0, 1, 2, 3, \dots\}$ ^{1 2} and the booleans $\text{Bool} = \{\text{true}, \text{false}\}$ even though we will only introduce them later. We will also use some basic operations such as addition ($_ + _$) on natural numbers.

Let's define a function which adds 2 to its input

```
add2 : ℕ → ℕ
add2 x = x + 2
```

To define `add2` we introduce a parameter x . Now we can apply `add2` to a number as in `3` and observe `add2 3` which is (no surprise) `5`. Lets do this step by step:

<code>add2 3 = 3 + 2</code>	Use the definition of <code>add2</code>
<code>= 5</code>	Calculating the sum.

¹In computer science unlike in mathematics we start counting with 0, which enables us to answer questions like *How many elephants are in your fridge?*

²You may think I have succumbed to set theory already by using $\{\dots\}$. But this is not the case, this is just a convenient notation for labelled coproducts which we will introduce soon.

In the first step we replace the parameter x with the actual argument 2 and then we use our knowledge about addition to conclude. The first step, replacing the parameter with the argument, is called β -reduction.

The definition above combines naming and defining the function. If we just want to write a function without giving it a name we can use λ -abstraction. That is we could have defined the same function as follows:

$$\begin{aligned} \text{add2} &: \mathbb{N} \rightarrow \mathbb{N} \\ \text{add2} &= \lambda x \rightarrow x + 2 \end{aligned}$$


We can view the first definition as an abbreviation for the 2nd. We can now give a more detailed derivation where we differentiate the unfolding of a definition and β -reduction.

$$\begin{aligned} \text{add2 } 3 &= (\lambda x \rightarrow x + 2) 3 && \text{Unfolding the definition of add2.} \\ &= 3 + 2 && \beta\text{-reduction} \\ &= 5 && \text{Calculating the sum.} \end{aligned}$$


Using λ we can only define a functions with one parameter. To define a function with several parameters we use *currying*, that is a function that returns a function

$$\begin{aligned} \text{add} &: \mathbb{N} \rightarrow (\mathbb{N} \rightarrow \mathbb{N}) \\ \text{add} &= \lambda x \rightarrow (\lambda y \rightarrow x + y) \end{aligned}$$

So $\text{add } 3 : \mathbb{N} \rightarrow \mathbb{N}$ is the function $\lambda y \rightarrow 3 + y$ that is the function that adds 3, and $(\text{add } 3) 5$ reduces to $5 + 3$ and hence 8. We can illustrate this – add outputs another box add 3:



which we can then feed with the 2nd argument to get the final output:



There are some syntactic conventions to simplify the use of currying: \rightarrow is *right associative*, hence we can write $\mathbb{N} \rightarrow \mathbb{N} \rightarrow \mathbb{N}$ for $\mathbb{N} \rightarrow (\mathbb{N} \rightarrow \mathbb{N})$ and application is *left associative* hence we can write $g\ 3\ 5$ for $(g\ 3)\ 5$. Moreover we can combine several λ -abstractions and write $\lambda\ x\ y\ \rightarrow\ x + y$ for $\lambda\ x\ \rightarrow\ (\lambda\ y\ \rightarrow\ x + y)$. Consequently the same convention also applies for the explicit function definition:

```
add : N → N → N
add x y = x + y
```

Instead of returning a function we can also have functions that have functions as input (these are called *higher order functions*). An example is:

```
k : (N → N) → N
k h = h 2 + h 3
```

Note that in this case we cannot omit the brackets in the function type. k is a function that gets a function on the natural numbers as input and returns a number. What is $k\ add2$? We can reason as follows:

$$\begin{aligned} k\ add2 &= add2\ 2 + add2\ 3 \\ &= (2 + 2) + (3 + 2) \\ &= 9 \end{aligned}$$

2.2 Identity and composition

Some functions work for any type, we call them *polymorphic*. An example is the identity function $id : A \rightarrow A$ which works for every type $A : \text{Set}$. In Agda we can write :

```
id : {A : Set} → A → A
id x = x
```

Writing $\{A : \text{Set}\} \rightarrow \dots$ indicates that the function works for every Set . Agda will automatically instantiate A that is we can just write $id\ 3 : \mathbb{N}$ and Agda infers that $A = \mathbb{N}$ in this case. In some cases it may be necessary to instantiate

the type variable explicitly in this case we can write $\text{id } \{\mathbb{N}\}$ or in case there are many parameters but we only want to instantiate a specific one we write $\text{id } \{A = \mathbb{N}\}$.

We will later explain types like $\{A : \text{Set}\} \rightarrow \dots$ which are actually instances of dependent types. In the moment we view polymorphism as a metamathematical notion, and we will use $A B C : \text{Set}$ to indicate variable types. In Agda this can be achieved by declaring:


```
variable
  A B C : Set
```

which means that I can write


```
id : A → A
id x = x
```

and Agda will automatically translate the type into $\{A : \text{Set}\} \rightarrow A \rightarrow A$.

We can draw a picture for the identity function which just consists of a wire from the input to the output:



Another example is function composition: given $g : A \rightarrow B$ and $f : B \rightarrow C$ we can construct a new function $f \circ g : A \rightarrow C$ which feeds the output of g into the input of f :



Composition can also be defined as a polymorphic function:

$$\begin{aligned} _ \circ _ &: (B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow (A \rightarrow C) \\ (f \circ g) x &= f (g x) \end{aligned}$$

In Agda we can define infix functions (and indeed more complicated syntactic schemes) by using `_` to indicate where the arguments go. The price we have to pay for this flexibility is that we have to separate any syntactical component with spaces, e.g. Agda would read `fog` just as one identifier.

Let's do an example. Earlier we define the function `add2 : ℕ → ℕ`, in the same vein we can define a squaring function:

```
square : ℕ → ℕ
square x = x * x
```

Now we can construct `square ∘ add2 : ℕ → ℕ`, what is `(square ∘ add2) 3`? Let's calculate:

$$\begin{aligned} (\text{square} \circ \text{add2}) 3 &= \text{square} (\text{add2} 3) \\ &= \text{square} (3 + 2) \\ &= \text{square} 5 \\ &= 5 * 5 \\ &= 25 \end{aligned}$$

This example can also be used to show that composition is not commutative,

what is $(\text{add2} \circ \text{square}) 2$?

$$\begin{aligned} (\text{add2} \circ \text{square}) 3 &= \text{add2} (\text{square } 3) \\ &= \text{add2} (3 * 3) \\ &= \text{add2 } 9 \\ &= 2 + 9 \\ &= 11 \end{aligned}$$

Hence we can see that $\text{square} \circ \text{add2}$ and $\text{add2} \circ \text{square}$ are different functions. In any case it isn't always possible to turn around composition. Assume as given a function $\text{isEven} : \mathbb{N} \rightarrow \text{Bool}$ that returns `true` if the input is even but `false` otherwise (e.g. $\text{isEven } 3 = \text{false}$ but $\text{isEven } 6 = \text{true}$). Now we can form $\text{isEven} \circ \text{square} : \mathbb{N} \rightarrow \text{Bool}$ (which actually behaves the same way as isEven) but it doesn't make sense to form $\text{square} \circ \text{isEven}$ because the output of isEven is `Bool` and this doesn't match the input of square which is \mathbb{N} .

You may note that there is a strange twist in the order of arguments to $_ \circ _$ which is visible in its type $(B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow (A \rightarrow C)$. Why is it not $(A \rightarrow B) \rightarrow (B \rightarrow C) \rightarrow (A \rightarrow C)$? And indeed when we evaluate $f \circ g$ we first evaluate g and then f . However, the reason is that function application is already written *the wrong way around*, that is when we evaluate $\text{square} (\text{add2 } 3)$ we first evaluate $\text{add2 } 3$ and then $\text{square } 5$, which is actually counterintuitive for people who write and read from left to right. Maybe we should write function application postfix 3 add2 instead of $\text{add2 } 3$. However, it seems to be too late to change this convention. And since $(f \circ g) x = f (g x)$ it seems wiser not to change the order when defining composition.

2.3 λ -calculus

I don't want to give a fully formal introduction to λ -calculus at this point. We will do this later when we have developed enough type theory to do this in Agda itself.

As we have already discussed we can view the explicit definition of a function like

$$\begin{aligned} _ \circ _ &: (B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow (A \rightarrow C) \\ (f \circ g) x &= f (g x) \end{aligned}$$

is a shorthand for an explicit definition using λ -terms:

$$\begin{aligned} _ \circ _ &: (B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow (A \rightarrow C) \\ _ \circ _ &= \lambda f \rightarrow \lambda g \rightarrow \lambda x \rightarrow f (g x) \end{aligned}$$

And since unfolding definitions is standard practice in Mathematics and elsewhere we concentrate on what happens to λ -terms.

When we form a λ -term we are using variables. Hence beginners often ask *What is the type of a variable?* The answer is that we assume that we are given a type of any variable, this is often called the (typing-) context.

The other cases are application and λ -abstraction:

variable A variable x has the type we have assumed for it in the context.

application Given a term $M : A \rightarrow B$ and a term $N : A$ we can form a term $M N : B$.

abstraction Given a variable x ³ if from assuming $x : A$ (here we extend the context) we can show that $M : B$ then we can form $\lambda x \rightarrow M : A \rightarrow B$. I need to add that we will only assume one type for a variable, that is by assuming $x : A$ we are ignoring all earlier assumptions regarding x .

This covers the *pure* λ -terms which don't use constructions specific to datatypes like `N` or `Bool` (but we may assume that the types of standard functions like `_+_` or `isEven` are in the context).

I have already mentioned β -equality. By β -equality we mean that we can reduce an application of a λ -abstraction to an argument by substituting the variable with the argument, that is

$$(\lambda x \rightarrow M) N = M[x:=N]$$

where $M[x:=N]$ is M with all occurrences of x replaced by N .

Hang on, actually it is a bit more complicated. We certainly do not want to replace bound variables, that is $(\lambda x \rightarrow (\lambda x \rightarrow x)) 3$ should just be $\lambda x \rightarrow x$ and not $\lambda x \rightarrow 3$. We remedy this by saying that $M[x:=N]$ is M with all *free* occurrences of x replaced by N . An occurrence is free if it is not bound.

We are not done yet! Consider $\lambda x \rightarrow (\lambda y \rightarrow x + y)$ and let's say we have a variable y lying around. What is $(\lambda x \rightarrow (\lambda y \rightarrow x + y)) y$? If we mechanically replace x by y we obtain $\lambda y \rightarrow y + y$. This is wrong because the y to refers to the bound variable y and not to our top-level y . This is called variable capture and it has to be avoided.

But then what is $(\lambda x \rightarrow (\lambda y \rightarrow x + y)) y$? Here we introduce another equality, α -equivalence which says that bound variables can be consistently replaced, i.e. $\lambda y \rightarrow x + y = \lambda z \rightarrow x + z$. Using this we can avoid capture by replacing the bound variable y avoiding capture, that is $(\lambda x \rightarrow (\lambda y \rightarrow x + y)) y = \lambda z \rightarrow y + z$. Note that any variable but y would work here.

There is yet another equation, η -equality which is motivated by the idea of extensionality, that is that two functions which are equal when applied to the same argument should be considered equal. In pure λ -calculus this is obtained by adding the following equation: assume $M : A \rightarrow B$ and given a variable $x : A$ which does not appear free in M , then

$$\lambda x \rightarrow M x = M$$

³Note that here x is not the specific variable called x but a metavariable standing for any concrete variable.

Reading it from right to left it means that if you want to show that two functions M and N are equal it is enough to show that $M\ x = N\ x$ where x is a variable which does not appear in M or N because

$$\begin{aligned} M &= \lambda\ x \rightarrow M\ x \\ &= \lambda\ x \rightarrow N\ x \\ &= N \end{aligned}$$

As an example we use η -equality to show that $f \circ \text{id} = f$:

$$\begin{aligned} f \circ \text{id} &= \lambda\ x \rightarrow f(\text{id}\ x) \\ &= \lambda\ x \rightarrow f\ x \\ &= f \end{aligned}$$

2.4 Combinatory logic

The polymorphic functions from section 2.2, id and $_ \circ _$ are also called combinators. We are going to introduce two combinators in this section, called S and K , which are *functionally complete*. This means that every function which we can write in (pure) λ -calculus can be written just using these two combinators. This applies only two pure functions which do not refer to other datatypes (like \mathbb{N} or Bool), which would require additional combinators. We are going to show this by providing a translation from λ -terms into terms only using combinators, which means that we can eliminate all variables. In a way combinators are a form of functional machine code and they have indeed been used for compilation.

We start with K which introduces constant functions:

$$\begin{aligned} K &: A \rightarrow B \rightarrow A \\ K\ x\ y &= x \end{aligned}$$

So for example $K\ 2 : B \rightarrow \mathbb{N}$ is the function that will always return 2. Note that this function is still polymorphic, it works for any type B .

The other combinator is S which generalizes composition, given types $A\ B\ C$:

$$\begin{aligned} S &: (A \rightarrow B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow (A \rightarrow C) \\ S\ f\ g\ x &= f\ x\ (g\ x) \end{aligned}$$

S is similar to $_ \circ _$ but the first function gets to see not only the result of the 2nd but also the original input. So for example $S\ (\lambda\ x\ y \rightarrow x + y)\ (\lambda\ x \rightarrow 2 * x)$ is the same as $\lambda\ x \rightarrow x + 2 * x$.

We can derive id (which is usually written I in combinatory logic) from S and K . The idea is that we can derive it from S by replacing the first argument f with K this obtaining $S\ K = \lambda\ g\ x \rightarrow K\ x\ (g\ x) = \lambda\ g\ x \rightarrow x$. To get the identity we can substitute g with any function. The simplest choice is K , i.e. $\text{id} = S\ K\ K$. When implementing this in Agda we run into a bit of a technical problem, given a type A we would like to define:

```
id : A → A
id = S K K
```

Actually Agda is going to complain because it cannot infer the argument B for the 2nd K – indeed we can use any type here. The simplest choice is A and we need to tell Agda this explicitly:

```
id : A → A
id {A} = S K (K {B = A})
```

We will now show that every pure lambda term can be translated into combinatory logic. I will do this informally we will later use Agda itself to give formal proof of this theorem.

Now let's do composition as an example. We can write $_ \circ _$ as a pure λ -term

```
\_ \circ \_ = \lambda f g x → f (g x)
```

Actually for our purposes it is better to expand the shorthand for repeated λ -abstraction:

```
\_ \circ \_ = \lambda f → \lambda g → \lambda x → f (g x)
```

Our strategy is to eliminate the λ s from inside out, that is first we translate $\lambda x \rightarrow f (g x)$ into combinators. This term will still contain the variables f and g . However, next we abstract g and then f to obtain a term using only combinators that contains no variables.

The terms which appear during the translation are terms which have no λ -abstractions but may contain variables, applications and the combinators S and K . We are going to construct $\lambda x \rightarrow M$ by looking at each of the possible cases:

M=x In this case clearly $\lambda x \rightarrow x = \text{id}$ which we have already obtained as $S K K$.

M=y If the variable is different from x we have a function $\lambda x \rightarrow y$ which is a constant function, that is $K y$.

M = M₁ M₂ In this case we use the assumption that we know already how to translate $\lambda x \rightarrow M_1$ and $\lambda x \rightarrow M_2$ Now we have

$$\begin{aligned} \lambda x \rightarrow M &= \lambda x \rightarrow M_1 M_2 \\ &= S (\lambda x \rightarrow M_1) (\lambda x \rightarrow M_2) \end{aligned}$$

which explains the need for S .

M = K Clearly $\lambda x \rightarrow K$ is just a constant function returning K , hence $K K$.

M = S As above $\lambda x \rightarrow S$ is just $K S$.

Ok, let's apply the translation to our example $\lambda f \rightarrow \lambda g \rightarrow \lambda x \rightarrow f(g x)$. We start with

$$\begin{aligned} & \lambda x \rightarrow f(g x) \\ &= S (\lambda x \rightarrow f) (\lambda x \rightarrow g x) \\ &= S (K f) (S (\lambda x \rightarrow g) (\lambda x \rightarrow x)) \\ &= S (K f) (S (K g) \text{id}) \end{aligned}$$

Before proceeding let's introduce some optimisations to avoid truly gigantic terms. So for example we can see that $\lambda x \rightarrow g x$ is just g using η . Another optimisation is that if the variable x does not appear in M then $\lambda x \rightarrow M$ is just $K M$ and there is no need to go through M . Using these optimisations we translate $_ \circ _$ in one go

$$\begin{aligned} & \lambda f \rightarrow \lambda g \rightarrow \lambda x \rightarrow f(g x) \\ &= \lambda f \rightarrow \lambda g \rightarrow S (\lambda x \rightarrow f) (\lambda x \rightarrow g x) \\ &= \lambda f \rightarrow \lambda g \rightarrow S (K f) g \\ &= \lambda f \rightarrow S (K f) \\ &= S (\lambda f \rightarrow S) (\lambda f \rightarrow K f) \\ &= S (K S) K \end{aligned}$$

While this term is comparatively small it is still hard to understand. The point of combinator is rather to show that we can avoid all the complications involving variables and just get away with S and K instead.

2.5 Products and Sums

So far we have only encountered the function type. In this section we will get to know some more basic type formers:

products Written $A \times B$, and also called cartesian product. In basic Mathematics we use products to represent coordinate systems, like $(1, 1) : \mathbb{R} \times \mathbb{R}$.

sums In Agda we write $A \uplus B$, I prefer $A + B$ but this clashes with $+$ on numbers. They are also called disjoint unions and coproducts.

Actually lets start with sums which are less common in conventional Mathematics but more common in Type Theory and functional programming.

2.5.1 Sums

Sums are necessary to represent alternatives. We have one thing or another. To define sums we use the **data** constructor:

```

data _ $\uplus$ _ (A B : Set) : Set where
  inj1 : A → A  $\uplus$  B
  inj2 : B → A  $\uplus$  B

```

This says that an element of $A \uplus B$ is either $\text{inj}_1 a$ where $a : A$ or it is $\text{inj}_2 b$ where $b : B$.

A simple example is that if in a form you can either fill in your order number (`OrdNum`) or your customer reference `CustRef` we can represent this as `OrdNum \uplus CustRef`. Note that even if we use the same number, lets say 1704 as an order number a customer reference we can always sure which is meant because we will use $\text{inj}_1 1704$ if it is an order number and $\text{inj}_2 1704$ if it is a customer reference.

The symbol for sums in Agda is a bit strange: it is a combination of the symbol for set theoretic union \cup and $+$, this is related to the name for sums in set theory *disjoint union*. I prefer to just use $+$ but this clashes with the use for numbers.

How are \cup and \uplus related? Indeed there is no operation \cup on types. The reason is that \cup in an *intensional* operator whose meaning depends on the choice of representation. For example let $A = \{a, b\}$ and $B = \{0, 1, 2\}$ then

$$A \cup B = \{a, b, 0, 1, 2\}$$

$$A \uplus B = \{\text{inj}_1 a, \text{inj}_1 b, \text{inj}_2 0, \text{inj}_2 1, \text{inj}_2 2\}$$

However, we can view A just as the one possible representation of a type with 2 elements. Another choice would be $A' = \{0, 1\}$. What happens now with \cup and \uplus ?

$$A' \cup B = \{0, 1, 2\}$$

$$A' \uplus B = \{\text{inj}_1 0, \text{inj}_1 1, \text{inj}_2 0, \text{inj}_2 1, \text{inj}_2 2\}$$

$A' \cup B$ now has only 3 elements, while $A' \uplus B$ again has 5 elements, hence is just another representation of a type with 5 elements. Thus \uplus is insensitive to the representation of a type, this is a property which holds for all operations definable in Type Theory (and hence in Agda).

What can we do with \uplus ? A useful general combinator is `case` that performs case analysis. we can define it using *pattern matching*:

```

case : (A → C) → (B → C) → A  $\uplus$  B → C
case f g (inj1 a) = f a
case f g (inj2 b) = g b

```

`case` chooses between two functions, `f` knows how to handle `A` and `g` know how to handle `B`. Going back to our example with the order form (`OrdNum \uplus CustRef`): if we have a function to lookup an order using the order number `lookupOrdNum : OrdNum → Order` and a function `lookupCustRef : CustRef → Order` we can use `case` to combine them

```

case lookupOrdNum lookupCustRef : OrdNum  $\uplus$  CustRef → Order

```

2.5.2 Products

We define products as a special case of a *record type*:

```
record _×_ (A B : Set) : Set where
  field
    proj1 : A
    proj2 : B
```

that is we specify the projections, which are functions with the types:

```
proj1 : A × B → A
proj2 : A × B → B
```

We can construct elements of $A \times B$ by specifying what the result is when applying projections. This is called *copatternmatching*. For example we can use this to define the pairing constructor which in Agda is just the `_ , _`, that is one doesn't have to put brackets around the tuple.

```
_ , _ : A → B → A × B
proj1 (a , b) = a
proj2 (a , b) = b
```

We can derive the constructor automatically by adding the constructor keyword to the record definition:

```
record _×_ (A B : Set) : Set where
  constructor _ , _
  field
    proj1 : A
    proj2 : B
```

What can we do with products? We derive a function `curry` that turns a function from products into a curried function:

```
curry : (A × B → C) → (A → B → C)
curry f = λ a b → f (a , b)
```

We can also do the reverse and translate a curried function back into one on products:

```
uncurry : (A → B → C) → (A × B → C)
uncurry g = λ x → g (proj1 x) (proj2 x)
```

Indeed, the two functions are inverse to each other, that is `curry (uncurry g) = g` and `uncurry (curry f) = f`. The first equation follows from what we already

know:

$$\begin{aligned}
 & \text{curry (uncurry g)} \\
 &= \lambda a b \rightarrow \text{uncurry g (a , b)} && \text{Unfolding curry} \\
 &= \lambda a b \rightarrow g (\text{proj}_1 (a , b)) (\text{proj}_2 (a , b)) && \text{Unfolding uncurry} \\
 &= \lambda a b \rightarrow g a b && \text{Evaluating projections} \\
 &= g && \text{Use } \eta \text{ twice.}
 \end{aligned}$$

For the 2nd equation we need an η -equality for products which says that $(\text{proj}_1 x , \text{proj}_2 x) = x$. This is also called *surjective pairing*:

$$\begin{aligned}
 & \text{uncurry (curry f)} \\
 &= \lambda x \rightarrow \text{curry f (proj}_1 x) (\text{proj}_2 x) && \text{Unfolding uncurry} \\
 &= \lambda x \rightarrow f (\text{proj}_1 x , \text{proj}_2 x) && \text{Unfolding curry} \\
 &= \lambda x \rightarrow f x && \text{Surjective pairing} \\
 &= f && \eta \text{ for functions.}
 \end{aligned}$$

This shows that the type of functions over products and curried functions are equivalent. Most functional programmers prefer to work with curried functions.

2.5.3 Strong sums?

We can try to play a similar game for coproducts. First of all using a product we can uncurry case:

$$\begin{aligned}
 & \text{case-c : (A } \rightarrow \text{ C) } \times \text{ (B } \rightarrow \text{ C) } \rightarrow \text{A } \uplus \text{ B } \rightarrow \text{C} \\
 & \text{case-c = uncurry case}
 \end{aligned}$$

and hence the inverse of case is:

$$\begin{aligned}
 & \text{uncase : (A } \uplus \text{ B } \rightarrow \text{C) } \rightarrow \text{(A } \rightarrow \text{C) } \times \text{(B } \rightarrow \text{C)} \\
 & \text{uncase h = } (\lambda a \rightarrow h (\text{inj}_1 a)) , (\lambda b \rightarrow h (\text{inj}_2 b))
 \end{aligned}$$

we would hope that the functions are inverse to each other, that is that $\text{uncase (case-c f) = f}$ and $\text{case-c (uncase g) = g}$. The first one is we can show:

$$\begin{aligned}
 & \text{uncase (case f)} \\
 &= (\lambda a \rightarrow \text{case-c f (inj}_1 a)) , (\lambda b \rightarrow \text{case-c f (inj}_2 b)) && \text{Unfold uncase} \\
 &= (\lambda a \rightarrow \text{proj}_1 f a , \lambda b \rightarrow \text{proj}_2 f b , \lambda b \rightarrow \text{proj}_2 f b) && \text{Unfold case-c and case.} \\
 &= (\text{proj}_1 f , \text{proj}_2 f) && \eta \text{ twice.} \\
 &= f && \text{Surjective pairing}
 \end{aligned}$$

However, the 2nd equality doesn't hold in Agda. Actually we can prove it using propositional equality $_ \equiv _$ we are going to introduce later. The equality $_ = _$

we have been considering here when talking *about*⁴ λ -terms is called definitional equality.

Indeed the η -laws for functions and surjective pairing is already extending the notion of *definitional* equality, they are mainly introduced for convenience. Indeed, the corresponding laws for other types, eg. the natural numbers, would destroy decidability and hence not be very useful for a language like Agda. It is possible, even though not entirely straightforward, to add η -laws for $_+_$ but this is usually avoided, mainly because it would quickly lead to an exponential blow up of cases and hence be computationally too costly.

2.5.4 Finite Types

The operation $_ \times _$ we have introduced handles the binary case of products. We can use it several times to handle more components, as in $A \times B \times C$. In practice it is better to use records which have more meaningful field names than proj_1 etc.

However, one interesting case not covered by binary products is the nullary product or unit type. In Agda this is denoted as \top and can be defined as a record with no fields:

```
record  $\top$  : Set where
  constructor tt
```

It has one element $\text{tt} : \top$.

We can play the same game with sums and derive the empty sum, that is a sum with no injections:

```
data  $\perp$  : Set where
```

The empty sum is the empty type, it is a type with no elements. A version of `case` for the empty type is quite useful:

```
case $\perp$  : {A : Set}  $\rightarrow \perp \rightarrow A$ 
case $\perp$  ()
```

We don't need any cases for \perp Agda marks this by `()` which basically means this case analysis has been intentionally left empty. We observe that we get a function from the empty type into any type. This is ok, because we will never be able to run this function since there isn't an element in the empty type as the name suggests.

We can use the unit type \top and sums \uplus to construct some finite types. For example `Two : Set` is a type with 2 elements:

```
Two : Set
Two =  $\top \uplus \top$ 
```

⁴Within Type Theory we cannot talk about definitional equality $_ = _$ is part of the language itself.

Here are the elements:

```
zero2 : Two
zero2 = inj1 tt
one2 : Two
one2 = inj2 tt
```

We can add one more element to `Two` to construct `Three` a type with 3 elements:

```
Three : Set
Three = T ⊔ Two
zero3 : Three
zero3 = inj1 tt
one3 : Three
one3 = inj2 zero2
two3 : Three
two3 = inj2 one2
```

However, as already in the case of products in practice it is usually preferable to use **data** to define labelled sums: ⁵

```
data Two : Set where
  zero2 one2 : Two
data Three : Set where
  zero3 one3 two3 : Three
```

How many elements are in `Two ⊔ Three`? An example is `inj1 one2`. Ok there is the choice between `inj1` and `inj2`, if we choose `inj1` we then have a choice between 2 elements and if we choose `inj2` we have a choice of 3 elements, and hence we can choose out of $2 + 3 = 5$ elements.

<code>inj₁ zero₂</code>	<code>inj₁ one₂</code>	<code>inj₂ zero₃</code>	<code>inj₂ one₃</code>	<code>inj₂ two₃</code>
---	--	---	--	--

In general if `A` has n elements and `B` has m elements then `A ⊔ B` has $m + n$ elements.

How many elements are in `Two × Three`? A typical example is `one2, two3`. Here we construct a pair and we have 2 independent choices, for the first components we have 2 options and for the second we have 3 and since these choices are independent we have $2 \times 3 = 6$ elements.

<code>zero₂, zero₃</code>	<code>zero₂, one₃</code>	<code>zero₂, two₃</code>
<code>one₂, zero₃</code>	<code>one₂, one₃</code>	<code>one₂, two₃</code>

In general if `A` has n elements and `B` has m elements then `A × B` are $m \times n$ elements.

⁵Note that we can combine constructors with the same type in one line.

What about the function type? How many elements are in $\text{Two} \rightarrow \text{Three}$? Here is a typical example:

```
foo : Two → Three
foo zero2 = two3
foo one2 = one3
```

We can define any such function by pattern matching: we have 3 choices in each case hence $3 \times 3 = 3^2 = 9$ elements. Now unlike products and sums function types are not symmetric. How many elements are in $\text{Three} \rightarrow \text{Two}$? Here again a typical example:

```
bar : Three → Two
bar zero3 = zero2
bar one3 = one2
bar two3 = zero2
```

There are 3 cases because there are 3 elements in **three** and in each case we have a choice of 2, hence we have $2 \times 2 \times 2 = 2^3 = 8$ elements. From these examples we can see that in general if **A** has n elements and **B** has m elements then $\text{A} \rightarrow \text{B}$ has m^n elements.

Indeed, in Mathematics the function type is often written as an exponential, that is $\text{A} \rightarrow \text{B}$ is written as B^A .

Previously, we have shown that types $\text{A} \times \text{B} \rightarrow \text{C}$ and $\text{A} \rightarrow \text{B} \rightarrow \text{C}$ are equivalent (via *curry* and *uncurry*). Writing this in exponential notation we find an old friend from high school:

$$C^{A \times B} = (C^B)^A$$

Similarly, with strong coproducts we would have that $(\text{A} \rightarrow \text{C}) \times (\text{B} \rightarrow \text{C})$ and $\text{A} \uplus \text{B} \rightarrow \text{C}$ are equivalent (via *case-c* and *uncase*). If we write $+$ for \uplus we arrive at another well known equation:

$$C^A \times C^B = C^{A+B}$$

2.6 History

λ -calculus was invented by Alonzo Church who used it in papers about formal logic in the 1930ies, e.g. [Chu32]. The story goes that the use of λ arose by accident: Church used a graphical notation where he connected the binding and the usage of a variable. I imagine he wrote something like this:



But the typesetter (this was not only before latex but before computers) went back and said, *Sorry Professor, I can't do that. But this looks a bit like a λ . Is it ok if I use just this?*



I don't know whether this is actually true but it is a nice story. Church realised that one of the problems of his calculus was that he could define too many weird functions, including the so called *fix point combinator* which allows to compute the fixpoint of any function by general recursion. But what is the fix point of negation? To avoid this sort of problems Church also invented the typed λ -calculus.⁶ Church had two primitive types, ι (the greek letter *iota*) standing for individuals and o (the greek letter *omikron* but it is the same as an o) standing for propositions, motivated by his intended application to formal logic. In my presentation I am not considering any fixed set of base types but I view them as type variables.

The untyped λ -calculus is still useful as a universal programming language which is an alternative to Turing machines. Indeed when Alan Turing visited Church they realised that Turing machines and λ -calculus compute the same set of computable functions. The generalisation of this observation to all computational mechanisms is known as the Church-Turing thesis.

In many papers about typed λ -calculus one starts with the untyped λ -calculus and then introduces a type system on top, as a way of assigning types to untyped terms. I do not follow this view which I think is too syntactic and considers typed terms as primitive. We call this *intrinsic typing*. Yes, it is true if we have to implement λ -terms we will at some point think about untyped terms but we also need to think about strings, characters, bits etc and will not talk about these implementation details.

Combinatory logic was actually invented before λ -calculus by Moses Schönfinkel in the 1920ies⁷. Again his interest was in formal logic and he realised

⁶A reference here is to a paper from 1940: [Chu40]

⁷If you can read German, you can check out [Sch24].

that combinators are a nice way to avoid the complexities of variables. This was also the motivation of Haskell Curry who reinvented combinatory logic more or less in the form we know it now.⁸ Some people say that the idea of *currying* a function with several arguments which is named after Curry, should have been called *schönfinkeln* but this never caught on - I wonder why? A good overview over the history of λ -calculus and combinatory logic is [CH06].

To understand my presentation of products and sums better (also called coproducts) one needs to know about category theory. Category theory is basically an abstract form of algebra which was introduced by Saunders MacLane and others.⁹ Their motivation came mainly from pure Mathematics, algebraic topology, I believe. However, later it was realised that category theory is also extremely useful for theoretical computer science. One important observation was Joachim Lambek's observation that a fundamental notion in category theory, *cartesian closed categories*, corresponds exactly to typed λ -calculus. In the categorical account products are already built-in and function types are characterised by the curry-uncurry-equivalence. One of the beautiful features of category theory is the presence of a mirror (*duality*) and it turns out that the mirror image (the dual) of products are sums or coproducts as we have introduced which are also extremely useful.

At this point you may wonder what the mirror image of functions is? This is a longer story but the short answer is that there isn't one. If you assume that there is one, the theory collapses: all λ -terms are equal.

λ -calculus is important for programming, to be precise in *functional programming*. Indeed the 2nd oldest programming language, LISP (The oldest is FORTRAN) was developed by John McCarthy in the late 1950ies and was inspired by the (untyped) λ -calculus [McC60]. Later types were introduced, famously by Robin Milner in the form of ML which stands for *Metalanguage* as a component of the Edinburgh LCF project (*Logic for computable functions*) [GMM⁺78]. Nowadays languages like Haskell (named after Haskell curry)¹⁰ and CAML are modern instances of typed functional programming and Agda's syntax and mechanisms borrow a lot especially from Haskell.

2.7 Exercises

1. Given

$$\begin{aligned} \text{add3} &: \mathbb{N} \rightarrow \mathbb{N} \\ \text{add3 } x &= x + 3 \end{aligned}$$

And for any type A:

$$\begin{aligned} \text{tw} &: (A \rightarrow A) \rightarrow A \rightarrow A \\ \text{tw } f \ n &= f (f \ n) \end{aligned}$$

⁸An early reference is See [Cur30]

⁹MacLanes book *Categories for the working Mathematician* is still worth a look. [ML78]

¹⁰A good introduction to Haskell is [Hut16]

Consider tw tw add3 1 .

- (a) What is its type? Justify your answer.
 - (b) What is its value? Show every step of the evaluation.
2. Derive a term in combinatory logic (that is only using S and K) for tw using

$$\begin{aligned} \text{tw} &: (A \rightarrow A) \rightarrow A \rightarrow A \\ \text{tw} &= \lambda f \rightarrow \lambda n \rightarrow f(f n) \end{aligned}$$

3. In section 2.2 we defined identity and composition:

$$\begin{aligned} \text{id} &: A \rightarrow A \\ \text{id } x &= x \\ _ \circ _ &: (B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow (A \rightarrow C) \\ (f \circ g) x &= f(g x) \end{aligned}$$

Using β and η show that the following equations¹¹ hold definitionally:

$$\begin{aligned} f \circ \text{id} &= f \\ \text{id} \circ f &= f \\ f \circ (g \circ h) &= (f \circ g) \circ h \end{aligned}$$

4. Define an operation with the following type:

$$A \times : (B \rightarrow C) \rightarrow A \times B \rightarrow A \times C$$

and show that the following equations hold using β , η and surjective pairing:

$$\begin{aligned} A \times \text{id} &= \text{id} \\ A \times (f \circ g) &= (A \times f) \circ (A \times g) \end{aligned}$$

This shows that the operation $A \times _$ on types is (definitionally) a functor.

5. Find elements of the following types using only pure λ -terms and the combinators for sums and products (proj_1 , proj_2 , $_$, $_$, inj_1 , inj_2 , case).

$$\begin{aligned} x_0 &: (A \rightarrow B) \rightarrow (B \rightarrow C) \rightarrow (A \rightarrow C) \\ x_1 &: (A \rightarrow B) \rightarrow ((A \rightarrow C) \rightarrow C) \rightarrow ((B \rightarrow C) \rightarrow C) \\ x_2 &: A \times (B \uplus C) \rightarrow (A \times B) \uplus (A \times C) \\ x_3 &: (A \times B) \uplus (A \times C) \rightarrow A \times (B \uplus C) \end{aligned}$$

¹¹These are the equations of a category. We are establishing that types and functions form a category, definitionally.